Mathematica Project 5 Help

Lagrange Multipliers (Constrained Optimization)
(read Section 13.8 of our text, see instructor's notes, and watch short videos posted on the web page)

The method of Lagrange multipliers (named after Joseph Louis Lagrange, 1736 - 1813) is a strategy for finding the local maxima and minima of a function subject to equality constraints. For the case of only one constraint and only two variables, consider the optimization problem:

maximize or minimize f(x,y) (which is called the objective function)
subject to g(x,y)=0 (which is called the constraint)

Assuming that both f and g have continuous first partial derivatives and introducing a new variable λ called a Lagrange multiplier, we will study the Lagrange function (or Lagrangian or Lagrangian expression) defined by $\mathcal{L}(x,y,\lambda)=f(x,y)-\lambda g(x,y)$. This $\mathcal{L}(x,y,\lambda)$ has critical points where its partial derivatives are zero, that is, $\mathcal{L}_x = \mathcal{L}_y = \mathcal{L}_\lambda = 0$, or $\mathcal{L}_x = f_x - \lambda g_x = 0$, $\mathcal{L}_y = f_y - \lambda g_y = 0$, $\mathcal{L}_\lambda = g(x,y) = 0$. Summarizing, we get equations $\nabla f = \lambda \nabla g$ and g(x,y)=0. Solve them, and then evaluate f at the points satisfying the equations, and choose the smallest and largest values of f in this list.

Example. Suppose we want to find the maximum values of $f(x,y)=x^2y$ with the condition that the x and y coordinates lie on the circle around the origin with radius 3, that is, subject to the constraint $x^2+y^2-9=0$.

In *Mathematica*, let us find partial derivatives of f and g and solve equations discussed above:

```
f[x_, y_] := x^2 * y;
g[x_, y_] := x^2 + y^2 - 9;
Dfx = D[f[x, y], x]
Dfy = D[f[x, y], y]
Dgx = D[g[x, y], x]
Dgy = D[g[x, y], y]
```

$2\,x\,y$

x^2

$2\,x$

$2\,y$

Let us first solves simultaneously the equations $\nabla f = \lambda \nabla g$ and g(x,y)=0 (resulting in six critical points for $\mathcal{L}$). Then we evaluate the objective function f at these points:

```
Solve[{2 * x * y - lambda * (2 * x) == 0 && x^2 - lambda * (2 * y) == 0 && x^2 + y^2 - 9 == 0},
 {x, y, lambda}]
f[x, y] /. Solve[{2 * x * y - lambda * (2 * x) == 0 &&
     x^2 - lambda * (2 * y) == 0 && x^2 + y^2 - 9 == 0}, {x, y, lambda}]
```

```
(* critical points *)
{{x → 0, y → -3, lambda → 0}, {x → 0, y → 3, lambda → 0},
 {x → -√6, y → -√3, lambda → -√3}, {x → -√6, y → √3, lambda → √3},
 {x → √6, y → -√3, lambda → -√3}, {x → √6, y → √3, lambda → √3}}
(* values of f at the six critical points *)
```

$\{0, 0, -6\sqrt{3}, 6\sqrt{3}, -6\sqrt{3}, 6\sqrt{3}\}$

Note that $6\sqrt{3}$ is the largest value, and this is the global maximum of f subject to constraint g(x,y)=0. It is attained twice: at the points $(\pm\sqrt{6}, \sqrt{3})$.

The value $-6\sqrt{3}$ is the global minimum of f at the points $(\pm\sqrt{6}, -\sqrt{3})$.

One more observation: the point (0 , 3) is a local minimum of f and (0 , − 3) is a local maximum of f, as may be determined by consideration of the Hessian matrix of $\mathcal{L}$ (x,y,0).

Now let us take a look at the level curves of f and g and what is going on at the point of extremum $(\sqrt{6}, -\sqrt{3})$:

```
pt = {x -> Sqrt[6], y -> -Sqrt[3]};
pt1 = {x, y} /. pt;
pt2 = pt1 + {Dfx, Dfy} /. pt;
pt3 = pt1 + {Dgx, Dgy} /. pt;
Show[ContourPlot[f[x, y], {x, -10, 10}, {y, -10, 10},
  PlotLegends → Automatic, Contours → {-30, -20, -10, 10, 20, 30}],
 ContourPlot[g[x, y] == 0, {x, -10, 10}, {y, -10, 10}],
 Graphics[{{PointSize[.015], Point[{Sqrt[6], -Sqrt[3]}]},
   Text[Style["(√6,-√3)", Bold], pt1 + {1.2, 1.4}]}],
 Graphics[{{Arrowheads[0.03], Arrow[{pt1, pt2}]},
   Text[Style["∇f", Bold], pt2 + {0.25, 0.25}]}],
 Graphics[{{Arrowheads[0.03], Arrow[{pt1, pt3}]},
   Text[Style["∇g", Bold], pt3 + {-0.25, -0.25}]}]
]
```



Note that the gradients of f and g are parallel at the point $(\sqrt{6}, -\sqrt{3})$ since we have $\nabla f = \lambda \nabla g$ there (with $\lambda = -\sqrt{3}$) by the method of Lagrange multipliers.
The gradients of f and g are parallel at the other three points of extremum as well.

For f and g , functions of three variables x, y, z, the method of Lagrange multipliers looks similar, that is, we have:

maximize or minimize f(x,y,z)
subject to g(x,y,z)=0

and we solve the equations $\nabla f = \lambda \nabla g$ and g(x,y,z)=0. Evaluate f at the points satisfying the equations, and choose the smallest and/or largest values of f in this list.

The method of Lagrange multipliers for two constraints is given by:

maximize or minimize f(x,y,z)
subject to g(x,y,z)=0 and h(x,y,z)=0

To find the solution, we need to solve the equations: $\nabla f = \lambda \nabla g + \mu \nabla h$, g(x,y,z)=0 and h(x,y,z) =0. Here λ and μ are two Lagrange multipliers to take care of the two constraints.

(Find examples in Section 13.8 of our text.)